



```

% in the lexical entries for verbs). %
VERB:
  <> == WORD
  <syn cat> == verb.

% MOR_WORD is the top node in a morphological hierarchy. It defaults to %
% UNDEFINED. The next DATR equation requires the evaluation of syntactic %
% category to determine the form of a morphological word. The equation %
% beginning <mor tact> defines the morphotactics. If the system is queried %
% for the TAM of nouns (<mor noun tam>) or the case of verbs %
% (<mor verb case>) then the output will be undefined. %
% Otherwise, a well formed noun or verb will have the form defined by %
% <mor tact>. Of course, it is possible for a noun to realize TAM, as this %
% is what happens with verbal case. But the equation beginning with %
% <mor noun case> rules this out for the ordinary cases. (We have not %
% included modal case in this fragment.) %

MOR_WORD:
  <> == UNDEFINED
  <mor word> == "<mor "<syn cat>" >"
  <mor tact> == "<root>" "<mor>"
  <mor noun tam> == UNDEFINED
  <mor verb case> == UNDEFINED
  <mor noun> == <mor tact>
  <mor verb> == <mor tact>.

% MOR_NOUN inherits from MOR_WORD and defines case endings, as well as %
% verbal case endings. The correspondence with the ordinary cases is clear. %
% With the verbal allative, we take the form -kiiwa and inherit the relevant %
% TAM inflection from the dental conjugation. With the verbal dative we %
% inherit the inflected word forms of the verb marutha and do not need %
% to specify the conjugation, because this will be inherited from that verb. %
% Note the use of the quotes around "Marutha:<mor word>". The effect of %
% these is to state that the value of <mor word> is whatever that may be %
% for Marutha. Recall from MOR_WORD that <mor word> involves evaluation %
% of syntactic category. As "Marutha:<mor word>" is quoted this will be %
% the value for the verb Marutha (i.e. <syn cat> == verb, inherited from %
% VERB). This suggests that the quoted node construct may meet some of the %
% descriptive requirements for representing a situation in which a fully %
% fledged word is becoming an affix but still has some of the original %
% properties of the class from which it originates. We have not modelled %
% the correspondence between the ordinary cases and verb inflections as %
% discussed in Evans (1995:254-55). To do this we would replace the values %
% in the RHS with reference to separate nodes for the individual forms. %
% These would then be associated with different features for nouns and %
% verbs. We have not done this, as we wish to focus on the key issue for %
% deponency of the verbal cases. %

MOR_NOUN:
  <> == MOR_WORD
  <mor case nom> == -a
  <mor case loc> == -ya
  <mor case abl> == -na
  <mor case prop> == -wuru
  <mor case obl> == -ntha
  <mor case all> == -r
  <mor case gen> == -karra
  <mor case assoc> == -nurru
  <mor case orig> == -wan-

```

```

<mor case priv> == -warri
<mor case cons> == -ngarrba
<mor case ins> == -nguni
<mor case util> == -marra
<mor case verbal all> == -kiiwa "V_DENTAL:<mor>"
<mor case verbal dat> == - "Marutha:<mor word>"
<mor case verbal trans> == - "Mariija:<mor word>"
<mor case verbal abl> == -wula "V_DENTAL:<mor>"
<mor case verbal evit> == - "Waalutha:<mor word>"
<mor case verbal don> == -wu "V_PALATAL:<mor>"
<mor case verbal purp> == - "Janija:<mor word>".

% MOR_VERB inherits from MOR_WORD and defines the verb inflections          %
% separately from the thematic marking in the dental and palatal             %
% conjugations. The element <mor theme> is defined by the inflection class.  %
% We use one example inflection, the actual.                                  %

MOR_VERB:
  <> == MOR_WORD
  <mor tam act> == "<mor theme>" -a.

% The node V_DENTAL inherits from MOR_VERB and specifies the theme          %
% consonant -th.                                                            %

V_DENTAL:
  <> == MOR_VERB
  <mor theme> == -th.

% The node V_PALATAL inherits from MOR_VERB and specifies the theme        %
% consonant -j.                                                              %

V_PALATAL:
  <> == MOR_VERB
  <mor theme> == -j.

% Lexical entries.                                                            %

Malaa:
  <> == NOUN
  <root> == mala
  <gloss> == sea.

Marutha:
  <> == VERB
  <mor> == V_DENTAL
  <root> == maru
  <gloss> == put.

Mariija:
  <> == VERB
  <mor> == V_PALATAL
  <root> == marii
  <gloss> == be put.

Waalutha:
  <> == VERB
  <mor> == V_DENTAL
  <root> == waalu
  <gloss> == drive away.

```

Janija:

```
<> == VERB
<mor> == V_PALATAL
<root> == jani
<gloss> == look for.
```

#hide

```
UNDEFINED
NOUN
VERB
WORD
MOR_WORD
MOR_NOUN
MOR_VERB
V_DENTAL
V_PALATAL
MOR_CAT.
```

#show

```
<syn cat>
<gloss>
<mor word case nom>
<mor word case loc>
<mor word case abl>
<mor word case prop>
<mor word case obl>
<mor word case all>
<mor word case gen>
<mor word case assoc>
<mor word case orig>
<mor word case priv>
<mor word case cons>
<mor word case ins>
<mor word case util>
<mor word case verbal all tam act>
<mor word case verbal dat tam act>
<mor word case verbal trans tam act>
<mor word case verbal abl tam act>
<mor word case verbal evit tam act>
<mor word case verbal don tam act>
<mor word case verbal purp tam act>
<mor word tam act>.
```